THE AMERICAN SOCIETY OF MECHANICAL ENGINEERS
345 E. 47 St., New York, N.Y. 10017

85-GT-219

L. J. PRITCHARD
Senior Aerodynamicist

Williams International
2280 W. Maple Road
Walled Lake, MI 48088 USA
Member, ASME

An Eleven Parameter Axial Turbine Airfoil Geometry Model

The mathematical derivation, and FORTRAN code, of a comprehensive but easy to use geometry model for axial flow turbine nozzles and rotors is presented. To uniquely define an airfoil on a cylinder the aerodynamicist need only specify the number of blades, and at each radius of interest: the axial and tangential chord, throat, uncovered turning, leading and trailing edge radii, inlet and exit blade angles, and inlet wedge angle. Default values exist for six of these geometric variables, which proves useful when starting a design. Both the suction and the pressure surfaces are described entirely by analytical functions. Sample airfoils are included that demonstrate the effect of each parameter upon blade shape.

Introduction

The creation of turbine blades is not an unsolved engineering problem. But axial flow turbomachines are a common component of gas turbine engines, therefore, the aerodynamicist is always looking for a better way to shape airfoils. Any turbine geometry model must be comprehensive, easy to use, and lend itself to the rapid, interactive computation of blade surface velocities. A literature search uncovered a variety of methods for generating individual airfoils including wrapping a numerically-defined thickness distribution around a numerically-defined camberline [1], methods with analytically-defined camber lines and/or analytically-defined thickness distributions [2], the Joukowski transformation [3], inverse methods which start with the desired surface velocities [4], and single polynomial blade shapes [5]. Unfortunately, these methods either require a lot of data, need additional smoothing, control the camberline instead of controlling the surfaces, cannot handle every case, are time consuming, or the geometric parameters are hard to quantify. However, the author's experience with airfoil geometry led to the formulation of a new method having none of these drawbacks. Presented here is a simplified version of a Rapid Axial Turbine Design algorithm, RATD, that does not require a cumbersome amount of input, but still has enough degrees of freedom to be universally applicable. To uniquely define an airfoil cascade on a cylinder requires only the eleven meaningful blade parameters described in Figure 1, and the immediate result is a nozzle or a rotor with analytically defined surfaces.



Figure 1. The Eleven Independent Geometric Parameters.

Presented at the Gas Turbine Conference and Exhibit
Houston, Texas – March 18-21, 1985

Identification of all Blade Parameters

At least 25 airfoil parameters can be associated with any blade shape. And every aerodynamicist must sort out which of these variables are known with absolute certainty, which must be iterated upon, and which simply result from specifying something else more important. For example, since

$$s = 2\pi R/N_B \tag{1}$$

the turbine designer could specify the pitch and the number of blades, and let the design radius fall out. But it is more practical to specify the radius and the number of blades, and accept the resultant pitch. In Table I the most important airfoil quantities encountered during a turbine design are identified and organized. Eleven independent blade parameters are found to be necessary and sufficient for creating an airfoil, and Figure 1 shows those eleven variables that experience has proven to be the most useful to work with. This model can successfully create hub, mean, and tip airfoils for both nozzles (Figure 2) and rotors (Figure 3). Moreover, this method works equally well for HP or LP turbine stages, and also can accommodate free vortex or variable work turbines. Furthermore, if all the blade parameters are smooth functions of radius, as shown in Figure 4, interpolated blade parameters always result in acceptable interpolated airfoils. With the radial dependence of each geometric variable specified, the distance from the center line becomes the only independent variable. Thus only 31 numbers are needed to completely define a blade row from hub to tip.

Table I. List of Airfoil Parameters.

INDEPENDENT PARAMETERS	DEPENDENT PARAMETERS
– RADIUS	– PITCH
– AXIAL CHORD*	– EXIT WEDGE ANGLE
– TANGENTIAL CHORD*	– STAGGER ANGLE
– UNGUIDED TURNING*	– AREA
– INLET BLADE ANGLE	– MAXIMUM THICKNESS
– INLET WEDGE ANGLE	– CHORD
– LEADING EDGE RADIUS*	– ZWEIFEL LOADING COEFFICIENT
– EXIT BLADE ANGLE	– SOLIDITY
– TRAILING EDGE RADIUS*	– x_{cg}
– NUMBER OF BLADES	– y_{cg}
– THROAT*	– INLET BLOCKAGE
	– EXIT BLOCKAGE
	– CAMBER ANGLE
*DEFAULT VALUES AVAILABLE	– LIFT COEFFICIENT

E–4649

NOMENCLATURE

a = POLYNOMIAL COEFFICIENT

a_i = AREA OF A TRIANGULAR SECTION OF THE AIRFOIL

A = CROSS SECTIONAL AREA OF THE AIRFOIL

b = POLYNOMINAL COEFFICIENT

c = CHORD, POLYNOMIAL COEFFICIENT

c_t = TANGENTIAL CHORD

c_x = AXIAL CHORD

C_L = LIFT COEFFICIENT

d = POLYNOMIAL COEFFICIENT

N_B = NUMBER OF BLADES

o = THROAT

R_{LE} = RADIUS OF THE LEADING EDGE CIRCLE

R_{TE} = RADIUS OF THE TRAILING EDGE CIRCLE

R = RADIUS OF THE CYLINDER UPON WHICH THE AIRFOIL IS DEFINED

s = PITCH

t_{max} = MAXIMUM THICKNESS OF THE AIRFOIL

x = ABSCISSA

x_0 = X-COORDINATE OF THE CENTER OF A CIRCLE

x_{cg} = X-COORDINATE OF THE CENTER-OF-GRAVITY OF THE AIRFOIL

$\bar{x}_i$ = X-COORDINATE OF THE CENTER-OF-GRAVITY OF A TRIANGULAR SECTION

y = ORDINATE

y_0 = Y-COORDINATE OF THE CENTER OF A CIRCLE

y_{cg} = Y-COORDINATE OF THE CENTER-OF-GRAVITY OF THE AIRFOIL

$\bar{y}_i$ = Y-COORDINATE OF THE CENTER-OF-GRAVITY OF A TRIANGULAR SECTION

z = Z-COORDINATE OF A POLAR COORDINATE SYSTEM

β = LOCAL BLADE ANGLE

ϵ = HALF-WEDGE ANGLE

ζ = UNGUIDED TURNING

θ = POLAR ANGLE

θ_c = CAMBER ANGLE

λ = BLOCKAGE

ξ = STAGGER ANGLE

σ = SOLIDITY

ψ_z = INCOMPRESSIBLE ZWEIFEL LOADING COEFFICIENT

SUBSCRIPTS

in = INLET

out = EXIT

LE = LEADING EDGE

TE = TRAILING EDGE

1, 2, 3, 4, 5 SEE FIGURE 5



Figure 2. A Typical RATD Nozzle and Channel.



Figure 3. A Typical RATD Rotor Hub, Mean, and Tip.



Figure 4. A Typical Independent Parameter Radial Dependence.

Default Values for the Independent Blade Parameters

Of the 25 (or more) unique variables that can be associated with every airfoil, only eleven are independent parameters. Furthermore, only five variables need to be well known at the start of a design, because default values can be used for the other six. The axial chord, for example, must be carefully controlled by the aerodynamicist during the design of a turbine. But if the axial chord is initially unknown it can be estimated by assuming zero incidence, zero deviation, and requiring the incompressible Zweifel loading coefficient [6] to be 0.8. Thus, the geometry results directly from the aerodynamics. (In this case, the default value is triggered by setting the parameter to zero). The tangential chord, on the other hand, is established from examining the surface velocity distributions [7]. But a good first guess comes from specifying the percent maximum-thickness-to-chord, or the axial chord and stagger angle, or by requiring that the rate of change of suction surface blade angle with axial distance be constant (set c_t = 0.0, and see Appendix A). The unguided turning can be determined by requiring the suction surface second derivative to be continuous at the throat. Next, the designer may specify the required percent blockages instead of selecting leading edge or trailing edge radii. Finally, if the throat is not well known, its default value comes from the blocked throat-to-pitch rule. Although no default values exist for the inlet wedge angle, it doesn't take long for the aerodynamicist to build up a data base that establishes a reasonable range for this variable.

Locating the Five Key Points on an Airfoil Surface

Whether default values are used or not, the eleven independent parameters translate into five points and five slopes on the cylinder of a given radius. These five key points on the airfoil surface result from 1) locating the leading and trailing edge circles in space, 2) finding the suction and pressure surface tangency points, and 3) setting the throat. The five key points are shown in Figure 5 and can be computed from the eleven independent parameters using the following equations:

Point No. 1
(Suction Surface Trailing Edge Tangency Point, see Appendix B)

$$\beta_1 = \beta_{out} - \epsilon_{out} \tag{2}$$

$$x_1 = c_x - R_{TE} \bullet (1 + \sin\beta_1) \tag{3}$$

$$y_1 = R_{TE} \bullet \cos\beta_1 \tag{4}$$

Point No. 2
(Suction Surface Throat Point)

$$\beta_2 = \beta_{out} - \epsilon_{out} + \zeta \tag{5}$$

$$x_2 = c_x - R_{TE} + (o + R_{TE}) \bullet \sin\beta_2 \tag{6}$$

$$y_2 = 2\pi R/N_B - (o + R_{TE}) \bullet \cos\beta_2 \tag{7}$$

Point No. 3
(Suction Surface Leading Edge Tangency Point)

$$\beta_3 = \beta_{in} + \epsilon_{in} \tag{8}$$

$$x_3 = R_{LE} \bullet (1 - \sin\beta_3) \tag{9}$$

$$y_3 = c_t + R_{LE} \bullet \cos\beta_3 \tag{10}$$

Point No. 4
(Pressure Surface Leading Edge Tangency Point)

$$\beta_4 = \beta_{in} - \epsilon_{in} \tag{11}$$

$$x_4 = R_{LE} \bullet (1 + \sin\beta_4) \tag{12}$$

$$y_4 = c_t - R_{LE} \bullet \cos\beta_4 \tag{13}$$

Point No. 5
(Pressure Surface Trailing Edge Tangency Point)

$$\beta_5 = \beta_{out} + \epsilon_{out} \tag{14}$$

$$x_5 = c_x - R_{TE} \bullet (1 - \sin\beta_5) \tag{15}$$

$$y_5 = - R_{TE} \bullet \cos\beta_5 \tag{16}$$

NOTE: The exit half-wedge angle is a dependent variable that must be iterated upon to remove the suction surface throat point discontinuity. A good first guess at the exit half-wedge angle is one-half the unguided turning.



Figure 5. The Five Key Points and the Five Surface Functions.

Mathematical Function Selection for the Airfoil Surfaces

In order to complete the airfoil description in the RΘ-Z Plane, the five key points must be connected. And the most natural way to connect them is by five mathematical functions. Logical choices for three of these functions are a leading edge circle, a trailing edge circle, and a circular arc describing the uncovered suction surface past the throat. The suction surface from the leading edge tangency point to the throat, and the entire pressure surface of the airfoil in Figure 5 are both defined by third order polynomials. A variety of mathematical functions were tried instead of the cubic, but a cubic is the most straight forward because it provides four simple algebraic equations and four unknown coefficients, as shown in Appendix C. At the points where they meet, all these functions and their first derivatives are piecewise continuous. However, the second derivatives and curvatures, in general, are not.

The Impact of Blade Parameters Upon Airfoil Shape

After experimenting with various combinations of input, and various computational schemes, a computer model was devised that starts with eleven parameters, determines the five key points and slopes, and creates the five curves in space that analytically define the airfoil. This procedure is summarized in Figure 6. Figures 7 through 13 represent the results of systematically varying seven of the eleven independent geometric parameters. Plotted are studies of the axial chord, tangential chord, unguided turning, inlet 1/2 wedge angle, throat, leading edge radius and trailing edge radius.



Figure 6. RATD Axial Turbine Design Flow Chart.



Figure 7. Axial Chord Study.



Figure 8. Tangential Chord Study.



Figure 9. Unguided Turn Study.



Figure 10. Inlet 1/2 – Wedge Angle Study.



Figure 11. Throat Study.



Figure 12. Leading Edge Radius Study.



Figure 13. Trailing Edge Radius Study.

List of Refinements

RATD, of course, is not the last word in geometry models. In its final form, any geometry model should include the capabilities listed in Table II. Most of these updates were incorporated in later versions of the geometry model presented in this paper. However, it is possible to create unacceptable airfoils with any model, and the one weakness of this model is an occasional flat spot (inflection point) on the suction surface near the leading edge.

Table II. List of Additional Geometry Model Requirements.

- ALTERNATE SURFACE FUNCTIONS TO CHOOSE FROM
- ELLIPTICAL LEADING EDGE
- PLOT OF BLADE ANGLE AND CURVATURE
- COMPUTATION OF MECHANICAL PROPERTIES (SEE APPENDIX D)
- RADIAL INTERPOLATION OF BLADE PARAMETERS
- LINKS TO FLOW PROGRAM, A BOUNDARY LAYER MODEL, AND A LOSS MODEL
- INTERACTIVE ON THE TEKTRONIX CRT
- ROTATE AIRFOIL
- STACK ON CG, LEADING EDGE, TRAILING EDGE, OR WITH COMPOUND LEAN
- MIRROR IMAGE
- CUT FLAT SECTIONS
- CREATE INSPECTION SHEET
- CREATE FINITE-ELEMENT INPUT FILE
- DESIGN ON CONES
- PLOT OF CHANNEL CONVERGENCE

E-4648

Summary

The creation of axial flow turbines has been boiled down to the basics without throwing away anything important. After analyzing all possible airfoil parameters, and synthesizing the results into a workable model, the "levers" that a turbine designer must "pull" are now well defined. To mathematically describe an airfoil, the aerodynamicist need specify only the parameters that are of fundamental interest. Furthermore, with the geometry quantified, it is useful to have a format that includes a numerical record of the geometry on each airfoil plot, including the dependent variables listed in Appendix E. Since each axial flow turbine rotor or nozzle is totally defined by eleven numbers, there is potential for standardization. This method has been successfully used by the author to design both axial flow turbine nozzles and rotors and can duplicate blade shapes created by other means. The more accessible the algorithm the more useful it is, and one of the strengths of this method is the compactness of its logic and computer code. For that reason a FORTRAN listing appears in Appendix F and a user's guide in Appendix G. The author would like to thank Gerry Large, Jim McKenna, and Ron Pampreen whose assistance made this geometry model possible.

Appendix A

The curve in Figure 14 goes through a point with a given slope, has the required slope at another value of x, and has constant turning in between. The value of y that goes with the second slope is also given.



Figure 14. Default Tangential Chord from $d\beta/dx$ = Constant.

Appendix B

Given the center of a circle, its radius, and the slope of a tangent line, what are the x,y coordinates of the tangency point? See Figure 15. NOTE: If the slope of the tangent line is negative (or positive) then so is beta.



Figure 15. Leading Edge and Trailing Edge Tangency Points.

Appendix C

Figure 16 shows the coefficients of the third order polynomial that goes through two points with the slopes given at those points.



Figure 16. A Cubic Determined From Two Points and Two Slopes.

Appendix D

The mechanical properties of an airfoil can be determined by dividing that airfoil into triangular pieces. Thus, given the vertices of a triangle, find its area and its center-of-gravity. See Figure 17.



Figure 17. The Mechanical Properties of a Triangle.

Appendix E

For reference, the dependent parameters defined by closed-form analytical expressions are listed here [8].

$$s = 2\pi R/N_B \tag{17}$$

$$\xi = \tan^{-1}\left(\frac{c_t}{c_x}\right) \tag{18}$$

$$A = \Sigma\ a_i \tag{19}$$

$$c = \sqrt{c_x^2 + c_t^2} \tag{20}$$

$$\psi_z = \frac{4\pi R}{c_x N_B} \bullet \sin(\beta_{in} - \beta_{out}) \bullet \frac{\cos\beta_{out}}{\cos\beta_{in}} \tag{21}$$

$$\sigma = c/s \tag{22}$$

$$x_{cg} = \frac{\Sigma\ \bar{x}_i\ a_i}{A} \tag{23}$$

$$y_{cg} = \frac{\Sigma\ \bar{y}_i\ a_i}{A} \tag{24}$$

$$\lambda_{in} = \frac{2R_{LE}}{s \bullet \cos\beta_{in}} \bullet 100 \tag{25}$$

$$\lambda_{out} = \frac{2R_{TE}}{s \bullet \cos\beta_{out}} \bullet 100 \tag{26}$$

$$\theta_c = \beta_{in} - \beta_{out} \tag{27}$$

$$C_L \sim \frac{2s}{c} \bullet \tfrac{1}{2}(\cos\beta_{in} + \cos\beta_{out}) \bullet (\tan\beta_{in} - \tan\beta_{out}) \tag{28}$$

$$\epsilon_{out} \sim \tfrac{1}{2}\zeta \tag{29}$$

Appendix F

Here is a FORTRAN 77 listing of the 327 line axial turbine geometry model, which uses Calcomp graphics, and runs on the Harris 800 computer.

```
      PROGRAM RATD
C...RAPID AXIAL TURBINE DESIGN PROGRAM.  WRITTEN JANUARY OF 1984 BY LJP
C...INPUT FILE = 25, OUTPUT FILE = 16.  SOME DEBUG WRITES COMMENTED OUT.
      DIMENSION XS(52),YS(52),XP(52),YP(52),TITLE(9),XC(24),YC(24)
      REAL NB
      LOGICAL ITER                      USED FOR TMAX/CHORD*100 ITERATION
      ICG=0
      PI=3.14159
      CALL PLOTS(0.,0.,-20)
      CALL PLOT(0.,-15.,-3)
      CALL PLOT(1.,2.5,-3)
      READ(25,8) TITLE
    8 FORMAT(9A6)
      READ(25,*) N,NB,FACTOR,ITRIG
      DO 1 II=1,N
      ITER=.FALSE.
      READ(25,*) R,CX,CT,THROAT,UGT,BIN,WEDGIN,RLE,BOUT,RTE
C...DEFAULT VALUES
      IF(RLE.GE.2.0) RLE=RLE/100.*2.*PI*R/NB*COS(BIN*PI/180.)/2.
      IF(RTE.GE.2.0) RTE=RTE/100.*2.*PI*R/NB*COS(BOUT*PI/180.)/2.
      IF(CX.LE.0.) CX=4.*PI*R/.8/NB*SIN((BIN-BOUT)*PI/180.)
     +*COS(BOUT*PI/180.)/COS(BIN*PI/180.)
      IF(THROAT.LE.0.) THROAT=2.*PI*R/NB*COS(BOUT*PI/180.)-2.*RTE
      IF(UGT.LE.0.) UGT=.0001
      WEDGOUT=UGT/2.
      IF(CT.LT.20.) GO TO 30
      CT=CX*TAN(CT*PI/180.)
      GO TO 27
C...IF CT IS BETWEEN 4.0 AND 20.0, THICKNESS/CHORD*100 IS ITERATED UPON
   30 IF(CT.GE.4.) ITER=.TRUE.
      IF(CT.GE.4.) TTC=CT/100.
      IF(CT.GE.4.) CT=0.
C...HERE THE KEY POINTS OF INTEREST ALONG THE BLADE ARE COMPUTED
   27 BETA1=BOUT-WEDGOUT
      X1=CX-RTE*(1.+SIN(BETA1*PI/180.))
      Y1=RTE*COS(BETA1*PI/180.)
      BETA2=BOUT-WEDGOUT+UGT
      X2=CX-RTE+(THROAT+RTE)*SIN(BETA2*PI/180.)
      Y2=2.*PI*R/NB-(THROAT+RTE)*COS(BETA2*PI/180.)
      BETA3=BIN+WEDGIN
      IF(CT.EQ.0.) CT=Y2+180./PI*(X2-X3)/(BETA2-BETA3)*ALOG
     +(COS(BETA2*PI/180.)/COS(BETA3*PI/180.))-RLE*COS(BETA3*PI/180.)
      X3=RLE*(1.-SIN(BETA3*PI/180.))
   20 Y3=CT+RLE*COS(BETA3*PI/180.)
      BETA4=BIN-WEDGIN
      X4=RLE*(SIN(BETA4*PI/180.)+1.)
      Y4=CT-RLE*COS(BETA4*PI/180.)
      BETA5=BOUT+WEDGOUT
      X5=CX-RTE*(1.-SIN(BETA5*PI/180.))
      Y5=-RTE*COS(BETA5*PI/180.)
      X6=CX
      Y6=0.
      X7=CX-RTE
      Y7=0.
      X8=0.
      Y8=CT
      X9=RLE
      Y9=CT
      AREA=0.
    2 FORMAT(F20.10)
C...THE CALCULATION OF THE UNCOVERED TURN CIRCLE
      X0=((Y1-Y2)*TAN(BETA1*PI/180.)*TAN(BETA2*PI/180.)
     ++X1*TAN(BETA2*PI/180.)-X2*TAN(BETA1*PI/180.))/(TAN(BETA2*PI/180.)
     +-TAN(BETA1*PI/180.))
      Y0=-(X0-X1)/TAN(BETA1*PI/180.)+Y1
      R0=SQRT((X1-X0)**2+(Y1-Y0)**2)
C...HERE THE EXIT WEDGE ANGLE IS ITERATED UPON TO REMOVE THROAT DISCONTINUITY
      YY2=Y0+SQRT(R0*R0-(X2-X0)**2)
C     WRITE(16,21) WEDGOUT,X2,Y2,YY2
      IF(ABS(Y2-YY2).LT..00001) GO TO 26     ! THROAT DISCONTINUITY REMOVED
      WEDGOUT=WEDGOUT*(Y2/YY2)**4
      IF(WEDGOUT.GT..001) GO TO 27           ! IF WEDGOUT<0 GO TO NEXT SECTION
      WRITE(16,28)
   28 FORMAT("0THE EXIT WEDGE ANGLE ITERATION FAILED.",/,
     +" THE EXIT WEDGE ANGLE WANTS TO GO NEGATIVE.",/,
     +" REDUCE THE EXIT BLADE ANGLE OR DECREASE THE THROAT.")
      GO TO 1
```

```
C...THE THIRD ORDER POLYNOMIAL CALL FOR SUCTION AND PRESSURE SURFACE
   26 CALL CUBIC(X2,Y2,BETA2,X3,Y3,BETA3,AS,BS,CS,DS)
      CALL CUBIC(X4,Y4,BETA4,X5,Y5,BETA5,AP,BP,CP,DP)
    3 FORMAT("1")
C...THIS IS THE SUCTION AND PRESSURE SURFACE DEFINITION / 50 POINTS PER SIDE
      XS(1)=X8
      YS(1)=Y8
      XP(1)=X8
      YP(1)=Y8
      DXP=(X4-X8)/9.
      DXS=(X3-X8)/9.
      DO 4 I=2,10
      XP(I)=XP(I-1)+DXP
      YP(I)=Y9-SQRT(RLE*RLE-(XP(I)-X9)**2)
      XS(I)=XS(I-1)+DXS
    4 YS(I)=Y9+SQRT(RLE*RLE-(XS(I)-X9)**2)
      DXP=(X5-X4)/30.
      DXS=(X2-X3)/20.
      DO 5 I=11,30
      XP(I)=XP(I-1)+DXP
      YP(I)=AP+XP(I)*(BP+XP(I)*(CP+XP(I)*DP))
      XS(I)=XS(I-1)+DXS
    5 YS(I)=AS+XS(I)*(BS+XS(I)*(CS+XS(I)*DS))
      DXS=(X1-X2)/10.
      DO 6 I=31,40
      XP(I)=XP(I-1)+DXP
      YP(I)=AP+XP(I)*(BP+XP(I)*(CP+XP(I)*DP))
      XS(I)=XS(I-1)+DXS
    6 YS(I)=Y0+SQRT(R0*R0-(XS(I)-X0)**2)
      DXP=(X6-X5)/10.
      DXS=(X6-X1)/10.
      DO 7 I=41,50
      XP(I)=XP(I-1)+DXP
      IF(XP(I).GT.CX) XP(I)=CX
      YP(I)=Y7-SQRT(RTE*RTE-(XP(I)-X7)**2)
      XS(I)=XS(I-1)+DXS
      IF(XS(I).GT.CX) XS(I)=CX
    7 YS(I)=Y7+SQRT(RTE*RTE-(XS(I)-X7)**2)
C..THE CALL TO THE MECHANICAL PROPERTIES SUBROUTINE
      CALL MECPRO(XS,YS,XP,YP,AREA,XCG,YCG)
C     WRITE(16,23) AREA,XCG,YCG
   23 FORMAT("0AREA, XCG, YCG=",3F10.5)
C...THE AIRFOIL MAXIMUM THICKNESS IS FOUND HERE
      TMAX=0.
      DO 17 I=1,50
      TM=999.
      DO 18 J=1,50
   18 TM=AMIN1(TM,SQRT((XS(I)-XP(J))**2+(YS(I)-YP(J))**2))
   17 TMAX=AMAX1(TMAX,TM)
      CHORD=SQRT(CT*CT+CX*CX)
C...MAX THICKNESS-TO-CHORD CONVERGENCE CRITERIA
      IF(.NOT.ITER) GO TO 19
      WRITE(16,21) CT,TMAX,CHORD,TTC
   21 FORMAT(" ",4F10.4)
      IF(ABS(TMAX/CHORD-TTC).LT..0001) GO TO 19
C...UPDATED TANGENTIAL CHORD ESTIMATE FOR T/C ITERATION
      CT=CT*(3.+TMAX/CHORD/TTC)/4.
      GO TO 20
C...AIRFOIL PLOTTING CODE
   19 XS(51)=0.
      YS(51)=0.
      XS(52)=1./FACTOR
      YS(52)=1./FACTOR
      XP(51)=0.
      YP(51)=0.
      XP(52)=1./FACTOR
      YP(52)=1./FACTOR
      IF(ITRIG.NE.4) CALL LINE(XS,YS,50,1,0,4)
      IF(ITRIG.NE.4) CALL LINE(XP,YP,50,1,0,4)
C...THE OUTPUT FILE IS CREATED HERE
      WRITE(16,11)
   11 FORMAT("1WILLIAMS INTERNATIONAL AXIAL TURBINE GEOMETRY MODEL",/,
     +"0    RADIUS    CX        CT        UGT    BETAIN  WEDGIN    RLE    BETAOUT
     +    RTE      THROAT     AREA      CGX      CGY")
      PITCH=2.*PI*R/NB
      STAG=ATAN(CT/CX)*180./PI
      ZWEIFEL=4.*PI*R/CX/NB*SIN((BIN-BOUT)*PI/180.)*COS(BOUT*PI/180.)
     +/COS(BIN*PI/180.)
      SOL=CHORD/PITCH
      PCTI=2.*RLE/COS(BIN*PI/180.)/PITCH*100.
      PCTE=2.*RTE/COS(BOUT*PI/180.)/PITCH*100.
      THETAC=BIN-BOUT
      CL=PITCH/CHORD*(COS(BIN*PI/180.)+COS(BOUT*PI/180.))*
```

```
     +(TAN(BIN*PI/180.)-TAN(BOUT*PI/180.))
      WRITE(16,15) R,CX,CT,UGT,BIN,WEDGIN,RLE,BOUT,RTE,THROAT,AREA,XCG
     +,YCG
   15 FORMAT("0",13F8.4)
C  31 WRITE(16,2) X0,Y0,R0,X1,Y1,BETA1,X2,Y2,BETA2,X3,Y3,BETA3,X4,Y4
C    +,BETA4,X5,Y5,BETA5,X6,Y6,X7,Y7,X8,Y8,X9,Y9,AS,BS,CS,DS,AP,BP,CP,DP
      WRITE(16,12)
   12 FORMAT("ONUMBER  SUCTION         PRESSURE",/,
     +"          X         Y         X         Y")
      DO 13 I=1,50
   13 WRITE (16,14) I,XS(I),YS(I),XP(I),YP(I)
   14 FORMAT(" ",I3,4F8.4)
C...CHECK FOR INFLECTION POINT ON PRESSURE SURFACE
      XINFL=-CP/3./DP
      IF(XINFL.GT.X4.AND.XINFL.LT.X5) WRITE(16,29)
   29 FORMAT("0THERE IS AN INFLECTION POINT IN THE PRESSURE SURFACE.",/,
     +" DECREASE UNGUIDED TURNING OR REDUCE TANGENTIAL CHORD.")
   16 FORMAT(" ")
C...CHANNEL PLOT
      DO 22 I=10,31
      SLOPE=BS+XS(I)*(2.*CS+3.*DS*XS(I))
      IF(SLOPE.NE.0.) S=-1./SLOPE
      ALPHA=ATAN(S)
      IF(SLOPE.EQ.0.) ALPHA=PI/2.
      XC(I-9)=XS(I)+THROAT*COS(ALPHA)
      YC(I-9)=YS(I)+THROAT*SIN(ALPHA)
      IF(S.LE.0.) XC(I-9)=XS(I)-THROAT*COS(ALPHA)
      IF(S.LE.0.) YC(I-9)=YS(I)-THROAT*SIN(ALPHA)
   22 CONTINUE
      XC(23)=0.
      YC(23)=0.
      XC(24)=1./FACTOR
      YC(24)=1./FACTOR
      IF(ITRIG.EQ.3) CALL LINE(XC,YC,22,1,0,4)
      IF(ITRIG.EQ.1) GO TO 9
C...PLOT OF BLADE STACKED ON CG
      IF(ITRIG.NE.4) GO TO 24
      DO 25 I=1,50
      XS(I)=XS(I)-XCG
      XP(I)=XP(I)-XCG
      YS(I)=YS(I)-YCG
   25 YP(I)=YP(I)-YCG
      IF(ICG.EQ.0) CALL PLOT(3.25,5.,-3)
      IF(ICG.EQ.0) CALL SYMBOL(0.,0.,.07,3,0.,-1)
      ICG=1
      CALL LINE(XS,YS,50,1,0,4)
      CALL LINE(XP,YP,50,1,0,4)
      GO TO 1
C...PLOT OF THE AIRFOIL ONE PITCH AWAY
   24 CALL SYMBOL(X2*FACTOR,Y2*FACTOR,.07,3,0.,-1)
      DO 10 I=1,50
      YP(I)=YP(I)+PITCH
   10 YS(I)=YS(I)+PITCH
      CALL LINE(XS,YS,50,1,0,4)
      CALL LINE(XP,YP,50,1,0,4)
    9 CALL TITLES(TITLE,R,CX,CT,THROAT,UGT,BIN,WEDGIN,RLE,BOUT,RTE,NB
     +,AREA,PITCH,STAG,FACTOR,TMAX,CHORD,ZWEIFEL,SOL,XCG,YCG,PCTI,PCTE
     +,WEDGOUT,THETAC,CL)
      CALL PLOT(8.5,0.,-3)
    1 CONTINUE
      IF(ITRIG.EQ.4) CALL PLOT(-3.25,0.,-3)
      CALL PLOT(7.5,0.,999)
      END
C
      SUBROUTINE CUBIC(X2,Y2,BETA2,X3,Y3,BETA3,A,B,C,D)
C...THIS SUBROUTINE PUTS A CUBIC THROUGH TWO POINTS WITH TWO SLOPES, AND
C...SOLVES FOR THE COEFFICIENTS OF Y=A+BX+CX**2+DX**3
      PI=3.14159
      D=(TAN(BETA2*PI/180.)+TAN(BETA3*PI/180.))/(X3-X2)/(X3-X2)
     +-2.*(Y2-Y3)/(X2-X3)/(X2-X3)/(X2-X3)
      C=(Y2-Y3)/(X2-X3)/(X2-X3)-TAN(BETA3*PI/180.)/(X2-X3)-D*(X2+2.*X3)
      B=TAN(BETA3*PI/180.)-2.*C*X3-3.*D*X3*X3
      A=Y3-B*X3-C*X3*X3-D*X3*X3*X3
      RETURN
      END
C
      SUBROUTINE TITLES(TITLE,R,CX,CT,THROAT,UGT,BIN,WEDGIN,RLE,BOUT,RTE
     +,NB,AREA,PITCH,STAG,FACTOR,TMAX,CHORD,ZWEIFEL,SOL,XCG,YCG,PCTI
     +,PCTE,WEDGOUT,THETAC,CL)
C...THIS SUBROUTINE LABELS AND PUTS LEGEND ON THE PLOT
      REAL NB
      INTEGER*3 DA(3),TI(3)
      CALL SYMBOL(0.,10.,.14,TITLE,0.,54)
      CALL SYMBOL(6.0,9.5,.07,2HR=,0.,2)
```

```
      CALL NUMBER(999.,999.,.07,R,0.,3)
      CALL SYMBOL(6.0,9.4,.07,3HCX=,0.,3)
      CALL NUMBER(999.,999.,.07,CX,0.,3)
      CALL SYMBOL(6.0,9.3,.07,3HCT=,0.,3)
      CALL NUMBER(999.,999.,.07,CT,0.,3)
      CALL SYMBOL(6.0,9.2,.07,4HUGT=,0.,4)
      CALL NUMBER(999.,999.,.07,UGT,0.,2)
      CALL SYMBOL(6.0,9.1,.07,8HBETA IN=,0.,8)
      CALL NUMBER(999.,999.,.07,BIN,0.,1)
      CALL SYMBOL(6.0,9.0,.07,14HHALF WEDGE IN=,0.,14)
      CALL NUMBER(999.,999.,.07,WEDGIN,0.,2)
      CALL SYMBOL(6.0,8.9,.07,4HRLE=,0.,4)
      CALL NUMBER(999.,999.,.07,RLE,0.,4)
      CALL SYMBOL(6.0,8.8,.07,9HBETA OUT=,0.,9)
      CALL NUMBER(999.,999.,.07,BOUT,0.,1)
      CALL SYMBOL(6.0,8.7,.07,4HRTE=,0.,4)
      CALL NUMBER(999.,999.,.07,RTE,.0,4)
      CALL SYMBOL(6.0,8.6,.07,3HNB=,0.,3)
      CALL NUMBER(999.,999.,.07,NB,0.,-1)
      CALL SYMBOL(6.0,8.5,.07,7HTHROAT=,0.,7)
      CALL NUMBER(999.,999.,.07,THROAT,0.,3)
      CALL SYMBOL(6.0,8.3,.07,6HPITCH=,0.,6)
      CALL NUMBER(999.,999.,.07,PITCH,0.,3)
      CALL SYMBOL(6.0,8.2,.07,15HHALF WEDGE OUT=,0.,15)
      CALL NUMBER(999.,999.,.07,WEDGOUT,0.,2)
      CALL SYMBOL(6.0,8.1,.07,8HSTAGGER=,0.,8)
      CALL NUMBER(999.,999.,.07,STAG,0.,1)
      CALL SYMBOL(6.0,8.0,.07,5HAREA=,0.,5)
      CALL NUMBER(999.,999.,.07,AREA,0.,4)
      CALL SYMBOL(6.0,7.9,.07,5HTMAX=,0.,5)
      CALL NUMBER(999.,999.,.07,TMAX,0.,4)
      CALL SYMBOL(6.0,7.8,.07,6HCHORD=,0.,6)
      CALL NUMBER(999.,999.,.07,CHORD,0.,3)
      CALL SYMBOL(6.0,7.7,.07,8HZWEIFEL=,0.,8)
      CALL NUMBER(999.,999.,.07,ZWEIFEL,0.,3)
      CALL SYMBOL(6.0,7.6,.07,9HSOLIDITY=,0.,9)
      CALL NUMBER(999.,999.,.07,SOL,0.,3)
      CALL SYMBOL(6.0,7.5,.07,4HXCG=,0.,4)
      CALL NUMBER(999.,999.,.07,XCG,0.,3)
      CALL SYMBOL(6.0,7.4,.07,4HYCG=,0.,4)
      CALL NUMBER(999.,999.,.07,YCG,0.,3)
      CALL SYMBOL(6.0,7.3,.07,12HINLET BLOCK=,0.,12)
      CALL NUMBER(999.,999.,.07,PCTI,0.,2)
      CALL SYMBOL(6.0,7.2,.07,11HEXIT BLOCK=,0.,11)
      CALL NUMBER(999.,999.,.07,PCTE,0.,2)
      CALL SYMBOL(6.0,7.1,.07,13HCAMBER ANGLE=,0.,13)
      CALL NUMBER(999.,999.,.07,THETAC,0.,1)
      CALL SYMBOL(6.0,7.0,.07,11HLIFT COEFF=,0.,11)
      CALL NUMBER(999.,999.,.07,CL,.0,2)
      CALL NUMBER(6.0,6.9,.07,FACTOR,0.,-1)
      CALL SYMBOL(999.,999.,.07,6HX SIZE,0.,6)
      CALL DATE(DA)
      CALL TIME(TI)
      CALL SYMBOL(5.93,6.7,.07,TI,0.,9)
      CALL SYMBOL(6.0,6.6,.07,DA,0.,9)
      RETURN
      END
C
      SUBROUTINE MECPRO(XS,YS,XP,YP,AREA,XCG,YCG)
C...SUBROUTINE TO COMPUTE CROSSECTIONAL AREA AND CENTER-OF-GRAVITY
C...BY BREAKING AIRFOIL UP INTO TRIANGLES
      DIMENSION XS(50),YS(50),XP(50),YP(50)
      A(X1,Y1,X2,Y2,X3,Y3)=ABS(X1*Y2+Y1*X3+Y3*X2-Y2*X3-Y1*X2-X1*Y3)/2.
      AREA=A(XS(1),YS(1),XS(2),YS(2),XP(2),YP(2))
      XCG=(XS(1)+XS(2)+XP(2))/3.*AREA
      YCG=(YS(1)+YS(2)+YP(2))/3.*AREA
      DO 1 I=2,48
      A1=A(XS(I),YS(I),XS(I+1),YS(I+1),XP(I),YP(I))
      AREA=AREA+A1
      XCG=XCG+(XS(I)+XS(I+1)+XP(I))/3.*A1
      YCG=YCG+(YS(I)+YS(I+1)+YP(I))/3.*A1
      A2=A(XS(I),YS(I),XP(I),YP(I),XP(I+1),YP(I+1))
      AREA=AREA+A2
      XCG=XCG+(XS(I)+XP(I)+XP(I+1))/3.*A2
    1 YCG=YCG+(YS(I)+YP(I)+YP(I+1))/3.*A2
      A1=A(XS(49),YS(49),XS(50),YS(50),XP(49),YP(49))
      AREA=AREA+A1
      XCG=XCG+(XS(49)+XS(50)+XP(49))/3.*A1
      YCG=YCG+(YS(49)+YS(50)+YP(49))/3.*A1
      XCG=XCG/AREA
      YCG=YCG/AREA
      RETURN
      END
```

Appendix G

There is a user's guide in Figure 18. Shown in Figure 19 are the input and the output it produces when run through the program of Appendix F.

TITLE									
N	NB	FACTOR	ITRIG						
R	CX	CT	THROAT	UGT	BIN	WEDGIN	RLE	BOUT	RTE

E-464

TITLE	Up to 54 characters
N	Number of airfoils to be generated
NB	Number of blades (real)
FACTOR	Plot scale factor
ITRIG	Plot trigger (1=1 blade, 2=2 blades, 3=2 blades + channel, 4 = C.G. stack)
R	Radius of airfoil design cylinder
CX	Axial chord (If CX = 0.0, ψ_z is set to 0.8)
CT	Tangential chord (If CT = 0.0, program assumes $d\beta/dx$ = constant) (If $4.0 \leqslant$ CT < 20.0, program interprets CT as t_{max}/chord x 100) (If CT $\geqslant 20.0$ program interprets CT as ξ)
THROAT	Throat (If throat = 0.0, throat set to s • cos BOUT − 2 • RTE)
UGT	Unguided turning
BIN	Inlet blade angle
WEDGIN	Inlet wedge angle
RLE	Leading edge radius (If RLE $\geqslant 2.0$, RLE is inlet percent blockage)
BOUT	Exit blade angle
RTE	Trailing edge radius (If RTE $\geqslant 2.0$, RTE is exit percent blockage)

ALL DATA IS FREE FORMAT.

AN UNLIMITED NUMBER OF SECTIONS CAN BE INPUT.

ALL ANGLES ARE IN DEGREES, AND IF THE SLOPE IS POSITIVE (OR NEGATIVE) THEN SO IS THE ANGLE.

ENGLISH OR METRIC DIMENSIONS CAN BE USED, BUT SOME UNITS MAY TRICK THE DEFAULT VALUES OF CT.

A-41856.1

Figure 18. RATD User's Guide.



Figure 19. Test Case Input and Output.

Bibliography

[1] Abbott and Von Doenhoff, Theory of Wing Sections, Dover Publications, New York, 1959.

[2] Dunham, J., "A Parametric Method of Turbine Blade Profile Design," ASME 74-GT-119.

[3] Spiegel, M., Complex Variables, McGraw-Hill, New York, 1964, p 224.

[4] Meauze, G., "An Inverse Time Marching Method for the Definition of Cascade Geometry," ASME 81-GT-167.

[5] Engeli, Zollinger, and Allemann, "A Computer Program for the Design of Turbomachinery Blades," ASME 78-GT-36.

[6] Sawyer, J. W., ed, Chapter 4, Sawyer's Gas Turbine Engineering Handbook, 2nd Edition, Vol I, Gas Turbine Publications, Inc., Stamford, Conn., 1972, p 70.

[7] Katsanis, T., "Fortran Program for Calculating Transonic Velocities on on a Blade-to-Blade Stream Surface of a Turbomachine," NASA TN D-5427, 1969.

[8] Horlock, J. H., Axial Flow Turbines, Krieger Publishing Company, Huntington, New York, 1973.